

NAG Library Routine Document

G05PCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05PCF generates a realization of a multivariate time series from a vector autoregressive moving average (VARMA) model. The realization may be continued or a new realization generated at subsequent calls to this routine.

2 Specification

```

SUBROUTINE G05PCF(MODE, K, XMEAN, IP, PHI, IQ, THETA, VAR, LDVAR, N, X,
1          IGEN, ISEED, R, NR, IWORK, LIWORK, IFAIL)

  INTEGER          MODE, K, IP, IQ, LDVAR, N, IGEN, ISEED(4), NR,
1          IWORK(LIWORK), LIWORK, IFAIL
  double precision XMEAN(K), PHI(IP*K*K), THETA(IQ*K*K), VAR(LDVAR,K),
1          X(LDVAR,N), R(NR)

```

3 Description

Let the vector $X_t = (x_{1t}, x_{2t}, \dots, x_{kt})^T$, denote a k dimensional time series which is assumed to follow a vector autoregressive moving average (VARMA) model of the form:

$$X_t - \mu = \phi_1(X_{t-1} - \mu) + \phi_2(X_{t-2} - \mu) + \dots + \phi_p(X_{t-p} - \mu) + \epsilon_t - \theta_1\epsilon_{t-1} - \theta_2\epsilon_{t-2} - \dots - \theta_q\epsilon_{t-q} \quad (1)$$

where $\epsilon_t = (\epsilon_{1t}, \epsilon_{2t}, \dots, \epsilon_{kt})^T$, is a vector of k residual series assumed to be Normally distributed with zero mean and positive-definite covariance matrix Σ . The components of ϵ_t are assumed to be uncorrelated at non-simultaneous lags. The ϕ_i 's and θ_j 's are k by k matrices of parameters. $\{\phi_i\}$, for $i = 1, 2, \dots, p$, are called the autoregressive (AR) parameter matrices, and $\{\theta_j\}$, for $j = 1, 2, \dots, q$, the moving average (MA) parameter matrices. The parameters in the model are thus the p k by k ϕ -matrices, the q k by k θ -matrices, the mean vector μ and the residual error covariance matrix Σ . Let

$$A(\phi) = \begin{bmatrix} \phi_1 & I & 0 & . & . & . & 0 \\ \phi_2 & 0 & I & 0 & . & . & 0 \\ . & & & . & & & \\ . & & & & . & & \\ \phi_{p-1} & 0 & . & . & . & 0 & I \\ \phi_p & 0 & . & . & . & 0 & 0 \end{bmatrix}_{pk \times pk} \quad \text{and} \quad B(\theta) = \begin{bmatrix} \theta_1 & I & 0 & . & . & . & 0 \\ \theta_2 & 0 & I & 0 & . & . & 0 \\ . & & & . & & & \\ . & & & & . & & \\ \theta_{q-1} & 0 & . & . & . & 0 & I \\ \theta_q & 0 & . & . & . & 0 & 0 \end{bmatrix}_{qk \times qk}$$

where I denotes the k by k identity matrix.

The model (1) must be both stationary and invertible. The model is said to be stationary if the eigenvalues of $A(\phi)$ lie inside the unit circle and invertible if the eigenvalues of $B(\theta)$ lie inside the unit circle.

For $k \geq 6$ the VARMA model (1) is recast into state space form and a realization of the state vector at time zero computed. For all other cases the routine computes a realization of the pre-observed vectors $X_0, X_{-1}, \dots, X_{1-p}, \epsilon_0, \epsilon_{-1}, \dots, \epsilon_{1-q}$, from equation (1), see Shea (1988). This realization is then used to generate a sequence of successive time series observations. Note that special action is taken for pure MA models, that is for $p = 0$.

At your request a new realization of the time series may be generated with less computation using only the information saved in a reference vector from a previous call to G05PCF. See the description of the parameter MODE in Section 5 for details.

The routine returns a realization of $X_1, X_2, \dots, X_n$. On a successful exit, the recent history is updated and saved in the array R so that G05PCF may be called again to generate a realization of $X_{n+1}, X_{n+2}, \dots$, etc. See the description of the parameter MODE in Section 5 for details.

Further computational details are given in Shea (1988). Note however that this routine uses a spectral decomposition rather than a Cholesky factorization to generate the multivariate Normals. Although this method involves more multiplications than the Cholesky factorization method and is thus slightly slower it is more stable when faced with ill-conditioned covariance matrices. A method of assigning the AR and MA coefficient matrices so that the stationarity and invertibility conditions are satisfied is described in Barone (1987).

One of the initialization routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05PCF.

4 References

Barone P (1987) A method for generating independent realisations of a multivariate normal stationary and invertible ARMA(p, q) process *J. Time Ser. Anal.* **8** 125–130

Shea B L (1988) A note on the generation of independent realisations of a vector autoregressive moving average process *J. Time Ser. Anal.* **9** 403–410

5 Parameters

1: MODE – INTEGER *Input*

On entry: a code for selecting the operation to be performed by the routine.

MODE = 0

Set up reference vector and compute a realization of the recent history.

MODE = 1

Generate terms in the time series using reference vector set up in a prior call to G05PCF.

MODE = 2

Combine the operations of MODE = 0 and 1.

MODE = 3

A new realization of the recent history is computed using information stored in the reference vector, and the following sequence of time series values are generated.

If MODE = 1 or 3, then you must ensure that the reference vector R and the values of K, IP, IQ, XMEAN, PHI, THETA, VAR and LDVAR have not been changed between calls to G05PCF.

Constraint: MODE = 0, 1, 2 or 3.

2: K – INTEGER *Input*

On entry: k , the dimension of the multivariate time series.

Constraint: $K \geq 1$.

3: XMEAN(K) – **double precision** array *Input*

On entry: μ , the vector of means of the multivariate time series.

4: IP – INTEGER *Input*

On entry: p , the number of autoregressive parameter matrices.

Constraint: $IP \geq 0$.

5: PHI(IP × K × K) – **double precision** array *Input*

On entry: must contain the elements of the $IP \times K \times K$ autoregressive parameter matrices of the model, $\phi_1, \phi_2, \dots, \phi_p$. If PHI is considered as a three-dimensional array, dimensioned as

PHI(K,K,IP), then the (i,j) th element of ϕ_l would be stored in PHI(i,j,l); that is, PHI($(l-1) \times k \times k + (j-1) \times k + i$) must be set equal to the (i,j) th element of ϕ_l , for $l = 1, 2, \dots, p$ and $i, j = 1, 2, \dots, k$.

Constraint: the elements of PHI must satisfy the stationarity condition.

- 6: IQ – INTEGER *Input*
On entry: q , the number of moving average parameter matrices.
Constraint: $IQ \geq 0$.
- 7: THETA(IQ $\times$ K $\times$ K) – **double precision** array *Input*
On entry: must contain the elements of the $IQ \times K \times K$ moving average parameter matrices of the model, $\theta_1, \theta_2, \dots, \theta_q$. If THETA is considered as a three-dimensional array, dimensioned as THETA(K,K,IQ), then the (i,j) th element of θ_l would be stored in THETA(i,j,l); that is, THETA($(l-1) \times k \times k + (j-1) \times k + i$) must be set equal to the (i,j) th element of θ_l , for $l = 1, 2, \dots, q$ and $i, j = 1, 2, \dots, k$.
Constraint: the elements of THETA must be within the invertibility region.
- 8: VAR(LDVAR,K) – **double precision** array *Input*
On entry: VAR(i,j) must contain the (i,j) th element of Σ . Only the lower triangle is required.
Constraint: the elements of VAR must be such that Σ is positive-definite.
- 9: LDVAR – INTEGER *Input*
On entry: the first dimension of the arrays VAR and X as declared in the (sub)program from which G05PCF is called.
Constraint: $LDVAR \geq K$.
- 10: N – INTEGER *Input*
On entry: n , the number of observations to be generated.
Constraint: $N \geq 0$.
- 11: X(LDVAR,N) – **double precision** array *Output*
On exit: X(i,t) will contain a realization of the i th component of X_t , for $i = 1, 2, \dots, k$ and $t = 1, 2, \dots, n$.
- 12: IGEN – INTEGER *Input*
On entry: must contain the identification number for the generator to be used to return a pseudorandom number and should remain unchanged following initialization by a prior call to G05KBF or G05KCF.
- 13: ISEED(4) – INTEGER array *Communication Array*
On entry: contains values which define the current state of the selected generator.
On exit: contains updated values defining the new state of the selected generator.
- 14: R(NR) – **double precision** array *Communication Array*
On entry: if MODE = 1 or 3, the array R as output from the previous call to G05PCF must be input without any change.
If MODE = 0 or 2, the contents of R need not be set.
On exit: information required for any subsequent calls to the routine with MODE = 1 or 3. See Section 8.

15: NR – INTEGER

Input

On entry: the dimension of the array R as declared in the (sub)program from which G05PCF is called.

Constraints:

$$\begin{aligned} &\text{if } K \geq 6, NR \geq (5r^2 + 1)K^2 + (4r + 3)K + 4; \\ &\text{if } K < 6, NR \geq ((IP + IQ)^2 + 1)K^2 + (4(IP + IQ) + 3)K + \\ &\quad \max\{Kr(Kr + 2), K^2(IP + IQ)^2 + l(l + 3) + K^2(IQ + 1)\} + 4. \end{aligned}$$

Where $r = \max(IP, IQ)$ and if $IP = 0$, $l = K(K + 1)/2$, or if $IP \geq 1$, $l = K(K + 1)/2 + (IP - 1)K^2$.

See Section 8 for some examples of the required size of the array R

16: IWORK(LIWORK) – INTEGER array

Workspace

17: LIWORK – INTEGER

Input

On entry: the dimension of the array IWORK as declared in the (sub)program from which G05PCF is called.

Constraint: $LIWORK \geq K \times \max(IP, IQ)$.

18: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 0$.

IFAIL = 2

On entry, $K < 1$.

IFAIL = 3

On entry, $IP < 0$.

IFAIL = 4

On entry, $IQ < 0$.

IFAIL = 5

On entry, $LDVAR < K$.

IFAIL = 6

On entry, $\text{MODE} \neq 0, 1, 2$ or 3.

IFAIL = 7

On entry, $\text{LIWORK} < K \times \max(\text{IP}, \text{IQ})$.

IFAIL = 8

On entry, NR is too small.

IFAIL = 9

On entry, the covariance matrix Σ , as stored in VAR, is not positive-definite.

IFAIL = 10

This is an unlikely exit brought about by an excessive number of iterations being needed by the NAG Library routine used to evaluate the eigenvalues of $A(\phi)$ or $B(\theta)$. If this error occurs please contact NAG.

IFAIL = 11

On entry, the autoregressive parameter matrices, as stored in PHI, are such that the model is non-stationary.

IFAIL = 12

On entry, the moving average parameter matrices, as stored in THETA, are such that the model is non-invertible.

IFAIL = 13

This is an unlikely exit brought about by an excessive number of iterations being needed by the NAG Library routine used to evaluate the eigenvalues of the covariance matrix.

IFAIL = 14

G05PCF has not been able to calculate all the required elements of the array R. This is likely to be because the AR parameters are very close to the boundary of the stationarity region.

IFAIL = 15

G05PCF has not been able to calculate all the required elements of the array R. This is an unlikely exit brought about by an excessive number of iterations being needed by the NAG Library routine used to evaluate eigenvalues to be stored in the array R. If this error occurs please contact NAG.

IFAIL = 16

Either R has been corrupted or the value of IP or IQ is not the same as when R was set up in a previous call with $\text{MODE} = 0$ or 2. To proceed, you should set MODE to 0 or 2.

7 Accuracy

The accuracy is limited by the matrix computations performed, and this is dependent on the condition of the parameter and covariance matrices.

8 Further Comments

Note that, in reference to IFAIL = 12, G05PCF will permit moving average parameters on the boundary of the invertibility region.

The elements of R contain amongst other information details of the spectral decompositions which are used to generate future multivariate Normals. Note that these eigenvectors may not be unique on different machines. For example the eigenvectors corresponding to multiple eigenvalues may be permuted. Although an effort is made to ensure that the eigenvectors have the same sign on all machines, differences in the signs may theoretically still occur.

The following table gives some examples of the required size of the array R, for $k = 1, 2$ or 3 , and for various values of p and q .

		q			
		0	1	2	3
p	0	13	20	31	46
		36	56	92	144
		85	124	199	310
	1	19	30	45	64
		52	88	140	208
		115	190	301	448
	2	35	50	69	92
		136	188	256	340
		397	508	655	838
	3	57	76	99	126
		268	336	420	520
		877	1024	1207	1426

Note that G13DXF may be used to check whether a VARMA model is stationary and invertible.

The time taken depends on the values of p , q and especially n and k .

9 Example

This program generates two realizations, each of length 48, from the bivariate AR(1) model

$$X_t - \mu = \phi_1(X_{t-1} - \mu) + \epsilon_t$$

with

$$\phi_1 = \begin{bmatrix} 0.80 & 0.07 \\ 0.00 & 0.58 \end{bmatrix},$$

$$\mu = \begin{bmatrix} 5.00 \\ 9.00 \end{bmatrix},$$

and

$$\Sigma = \begin{bmatrix} 2.97 & 0 \\ 0.64 & 5.38 \end{bmatrix}.$$

The pseudorandom number generator is initialized by a call to G05KBF. Then, in the first call to G05PCF, MODE is set to 2 in order to set up the reference vector before generating the first realization. In the subsequent call MODE is set to 3 and a new recent history is generated and used to generate the second realization.

9.1 Program Text

```

*      G05PCF Example Program Text
*      Mark 20 Release. NAG Copyright 2001.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          KMAX, LDVAR, IPMAX, IQMAX, NMAX, NR, LIWORK
      PARAMETER        (KMAX=3,LDVAR=KMAX,IPMAX=2,IQMAX=2,NMAX=100,
+                      NR=600,LIWORK=10)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, IGEN, II, IP, IQ, J, K, L, MODE, N
*      .. Local Arrays ..
      DOUBLE PRECISION PHI(KMAX*KMAX*IPMAX), R(NR),
+                      THETA(KMAX*KMAX*IQMAX), VAR(LDVAR,KMAX),
+                      X(LDVAR,NMAX), XMEAN(KMAX)
      INTEGER          ISEED(4), IWORK(LIWORK)
*      .. External Subroutines ..
      EXTERNAL         G05KBF, G05PCF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G05PCF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) K, IP, IQ, N
*
      IF (K.GT.0 .AND. K.LE.KMAX .AND. IP.GE.0 .AND. IP.LE.IPMAX .AND.
+      IQ.GE.0 .AND. IQ.LE.IQMAX) THEN
      IF (N.GT.0 .AND. N.LE.NMAX) THEN
        DO 40 L = 1, IP
          DO 20 I = 1, K
            II = (L-1)*K*K + I
            READ (NIN,*) (PHI(II+K*(J-1)),J=1,K)
20          CONTINUE
40          CONTINUE
          DO 80 L = 1, IQ
            DO 60 I = 1, K
              II = (L-1)*K*K + I
              READ (NIN,*) (THETA(II+K*(J-1)),J=1,K)
60          CONTINUE
80          CONTINUE
          READ (NIN,*) (XMEAN(I),I=1,K)
          DO 100 I = 1, K
            READ (NIN,*) (VAR(I,J),J=1,I)
100         CONTINUE
*      Initialize the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 4234401
      ISEED(4) = 742355
*      IGEN identifies the stream.
      IGEN = 1
      CALL G05KBF(IGEN,ISEED)
*
      MODE = 2
      IFAIL = 1
*
      CALL G05PCF(MODE,K,XMEAN,IP,PHI,IQ,THETA,VAR,LDVAR,N,X,IGEN,
+              ISEED,R,NR,IWORK,LIWORK,IFAIL)
*
      WRITE (NOUT,*)
      IF (IFAIL.EQ.0) THEN
*
        WRITE (NOUT,*) ' Realisation Number 1 '
*
        DO 120 I = 1, K
          WRITE (NOUT,99999) ' Series number ', I
          WRITE (NOUT,*) ' -----'
          WRITE (NOUT,*)
          WRITE (NOUT,99998) (X(I,J),J=1,N)
120       CONTINUE
*

```

```

MODE = 3
IFAIL = -1
*
CALL G05PCF(MODE,K,XMEAN,IP,PHI,IQ,THETA,VAR,LDVAR,N,X,
+          IGEN,ISEED,R,NR,IWORK,LIWORK,IFAIL)
*
IF (IFAIL.EQ.0) THEN
  WRITE (NOUT,*)
  WRITE (NOUT,*)
  WRITE (NOUT,*) ' Realisation Number 2'
*
  DO 140 I = 1, K
    WRITE (NOUT,99999) ' Series number ', I
    WRITE (NOUT,*) ' -----'
    WRITE (NOUT,*)
    WRITE (NOUT,99998) (X(I,J),J=1,N)
140    CONTINUE
  END IF
ELSE
  WRITE (NOUT,99997) IFAIL
END IF
*
END IF
END IF
*
99999 FORMAT (/1X,A,I3)
99998 FORMAT (8(2X,F8.3))
99997 FORMAT (1X,' ** G05PCF returned with IFAIL = ',I5)
END

```

9.2 Program Data

G05PCF Example Program Data

```

2 1 0 48          : K, IP, IQ, N
0.80 0.07
0.00 0.58          : End of matrix PHI(,,1)
5.00 9.00          : XMEAN
2.97
0.64 5.38          : End of matrix VAR

```

9.3 Program Results

G05PCF Example Program Results

Realisation Number 1

Series number 1

5.765	3.983	0.496	-0.754	-1.718	-2.753	-0.568	1.719
2.380	3.320	3.150	2.051	1.304	2.079	2.940	3.539
3.876	3.179	7.230	5.791	6.297	6.508	7.993	7.041
5.568	6.404	6.283	3.230	0.989	-0.397	-0.331	2.526
1.629	2.693	3.093	2.967	5.162	6.331	8.169	10.801
9.174	5.627	7.696	5.844	1.539	2.377	1.335	-0.144

Series number 2

14.749	13.246	8.826	8.614	8.779	7.310	5.961	4.594
5.312	6.354	6.843	7.136	10.653	9.416	10.083	11.012
10.745	8.807	4.318	6.509	7.520	9.205	9.877	8.082
8.408	11.218	9.648	9.260	6.631	10.407	7.784	6.509
5.174	8.594	7.900	8.575	8.307	9.726	5.303	9.989
9.924	13.597	16.134	15.464	11.816	8.428	11.221	7.869

Realisation Number 2

Series number 1

6.253	9.454	8.619	7.403	6.684	7.998	6.669	7.011
7.899	8.984	7.873	8.161	6.462	5.396	2.543	3.104
4.285	3.837	1.109	2.372	5.804	1.929	6.479	5.964
6.594	8.835	9.217	8.546	7.011	7.797	8.386	8.290
8.227	4.163	4.232	5.161	6.794	7.188	6.552	7.101
5.061	3.964	6.517	6.315	3.989	4.552	3.079	4.139

Series number 2

11.114	13.425	10.316	9.063	6.324	3.673	5.568	7.103
8.561	9.097	11.745	10.028	12.138	9.973	8.747	10.753
15.455	10.861	14.161	9.624	12.976	10.141	7.931	8.289
9.520	10.412	8.248	4.229	3.834	6.840	8.367	12.120
13.373	14.411	9.508	12.724	11.858	12.463	14.742	12.301
14.039	12.476	13.437	11.757	11.972	9.273	9.496	6.394